

初学者园地  
**手把手**  
**教你学单片机(十)**

[套件供应]

■周兴华

下面开始做实验,感性认识 MCS-51 单片机中定时/计数器的作用。

在 S1 板上做一个实验,使用定时器 T1 以方式 0 使单片机产生周期为 1000μS 等宽方波脉冲(1000Hz 音频),在 P1.7 输出驱动蜂鸣器发音。

S1 板使用 11.0592MHz 晶振,可近似认为其为 12MHz。这样一个机器周期为 1μS。欲产生 1000μS 周期方波脉冲,只需在 P1.7 以 500μS 时间交替输出高低电平即可。

1.T1 为方式 0,则 M1M0=00H。使用定时功能,C/T=0。GATE=0。TO 不用,其有关位设为 0。这样, TMOD=00H。

2. 方式 0 为 13 位长度计数结构,设计数初值为 X,则: $(2^{13}-X) \times 1 \times 10^{-6} = 500 \times 10^{-6}$  得 X=7692D

X=1111000001100B 转成 16 进制后,高 8 位=FOH,低 8 位=0CH。即 TH1=FOH,TL1=0CH。

3. 由控制寄存器 TCON 中的 TR1 位来控制定时的启动和停止, TR1=1 启动,TR1=0 停止。

在我的文档中建立一个文件目录(S14),然后建立一个 S14.uv2 的工程项目,最后建立源程序文件(S14.asm)。

输入下面的程序:

```

序号:1      ORG 0000H
2          LJMP MAIN
3          ORG 030H
4  MAIN:MOV TMOD,#00H
5          MOV TH1,#0F0H
6          MOV TL1,#0CH
7          MOV IE,#00H
8          SETB TR1
    
```

```

9  LOOP:JBC TF1,LOOP1
10         AJMP LOOP
11  LOOP1:MOV TH1,#0F0H
12         MOV TL1,#0CH
13         CLR TF1
14         CPL P1.7
15         AJMP LOOP
16         END
    
```

编译通过后,将 S14 文件夹中的 hex 文件烧录到 89C51 芯片中,将芯片插入到 S1 型 LED 试验板上,接上 5V 稳压电源,蜂鸣器中立即响起悦耳的 1KHz 音频声。

下面我们对程序进行分析解释。

序号 1(程序解释,以下同):程序开始。

序号 2:跳转到 MAIN 主程序处。

序号 3:主程序 MAIN 从地址 0030H 开始。

序号 4:置 T1 为方式 0。

序号 5、6:载入定时初值。

序号 7:禁止中断。

序号 8:启动定时器 T1。

序号 9、10:查询 T1 的溢出标志 TF1。TF1=0 定时未到,转 LOOP 继续查询;TF1=1 定时到,转 LOOP1。

序号 11、12:重装定时初值。

序号 13:清除溢出标志。

序号 14:P1.7 输出端取反。

序号 15:跳转到 LOOP 处重复循环。

序号 16:程序结束。

再在 S1 板上做一个实验,使用定时器 T1 以方式 2 计数,每计 10 次,进行累加器加 1 操作,并送 P1 口显示。

1.T1 为方式 2,则 M1M0=10H。使用计数功能,C/T=1。GATE=0。TO 不用,其有关位设为 0。这样,

TMOD=60H。

2.方式 2 为 8 位长度自动重装计数结构,设计数初值为: $(2^8-10) = 246D = 11110110B = F6H$ ,即 TH1=0F6H,TL1=0F6H。

3. 由控制寄存器 TCON 中的 TR1 位来控制定时的启动和停止, TR1=1 启动,TR1=0 停止。

在我的文档中建立一个文件目录(S15),然后建立一个 S15.uv2 的工程项目,最后建立源程序文件(S15.asm)。

输入下面的程序:

```

序号: 1      ORG 0000H
2          LJMP MAIN
3          ORG 030H
4  MAIN:MOV TMOD,#60H
5          MOV TH1,#0F6H
6          MOV TL1,#0F6H
7          MOV IE,#00H
8          SETB TR1
9          MOV P1,#00H
10         ACALL DEL
11  LOOP:JBC TF1,LOOP1
12         AJMP LOOP
13  LOOP1:INC A
14         MOV P1,A
15         ACALL DEL
16         AJMP LOOP
17  DEL:MOV R7,#014H
18  DEL1:MOV R6,#0FFH
19  DEL2:MOV R5,#01FH
20  DEL3:DJNZ R5,DEL3
21         DJNZ R6,DEL2
22         DJNZ R7,DEL1
23         RET
24         END
    
```

编译通过后,将 S15 文件夹中的 hex 文件烧录到 89C51 芯片中,将芯片插入到 S1 型 LED 试验板上,接上 5V 稳压电源,这时 P1 口外接的 8 个 LED 均点亮(输出状态为 00H)。将 S1 板配带的试验线一端插到标示有“0”电平的排针上,另一端去触碰标示为 P3.5(T1)的排针,可发现,每触碰 10 次后,P1 口按二进

制加 1(→01H→02H...)。有些读者可能会发现,触碰不到 10 次,P1 口也按二进制加 1。其实这是由于触碰时的抖动效应,可能一下输入了好几个脉冲。这丝毫也不影响我们对程序的理解。

下面我们对程序进行分析解释。

序号 1(程序解释,以下同):程序开始。

序号 2:跳转到 MAIN 主程序处。

序号 3:主程序 MAIN 从地址 0030H 开始。

序号 4:置 T1 为方式 2。

序号 5、6:载入定时初值。

序号 7:禁止中断。

序号 8:启动定时器 T1。

序号 9:点亮 P1 口的 LED。

序号 10:调用延时子程序,便于观察 LED 点亮状态。

序号 11、12:查询 T1 的溢出标志 TF1。TF1=0 定时未到,转 LOOP 继续查询;TF1=1 定时到,转 LOOP1。

序号 13:累加器加 1。

序号 14:累加器内容送 P1 口显示。

序号 15:调用延时子程序,便于观察 LED 点亮状态。

序号 16:跳转到 LOOP 处重复循环。

序号 17~23:延时子程序。

序号 24:程序结束。

最后再在 S1 板上做一个实验,使用定时器 T1 以方式 1 定时,使 P1.0 端每隔 1 分钟取反一次。

S1 板使用 11.0592MHz 晶振,可近似认为其为 12MHz。这样一个机器周期为 1μs。在方式 1 最大定时时间 =  $65536 \times 12 / (12 \times 106) = 65536 \mu s = 65.536 ms$ ,显然离 1 分钟还差十万八千里。这里我们将 T1 设定为定时 50ms,另设两个软件计数器,采用 30H、31H 两个单元进行秒、分计数。30H 内置常数 20,31H 内置常数 60,这样  $50ms \times 20 \times 60 = 60000ms = 60s = 1$  分钟。

1.T1 为方式 1,则 M1M0=01H。使用定时功能,C/T=0。GATE=0。TO

不用,其有关位设为 0。这样, TMOD=10H。

2.方式 1 为 16 位长度计数结构,计数初值为:  $(216-X) \times 1 \times 10^{-6} = 50 \times 10^{-3}$ ,  $X=65536-50000=15536D=3CB0H$ ,即 TH1=3CH,TL1=0B0H。

3.由控制寄存器 TCON 中的 TR1 位来控制定时的启动和停止,TR1=1 启动,TR1=0 停止。

在我的文档中建立一个文件目录(S16),然后建立一个 S16.uv2 的工程项目,最后建立源程序文件(S16.asm)。

输入下面的程序:

```

序号:1      ORG 0000H
2           LJMP MAIN
3           ORG 030H
4   MAIN:   MOV 30H,#20
5           MOV 31H,#60
6           MOV TMOD,#10H
7           MOV TH1,#3CH
8           MOV TL1,#0B0H
9           MOV IE,#00H
10          SETB TR1
11  LOOP:   JBC TF1,LOOP1
12          AJMP LOOP
13  LOOP1:  MOV TH1,#3CH
14          MOV TL1,#0B0H
15          DJNZ 30H,LOOP
16          MOV 30H,#20
17          DJNZ 31H,LOOP
18          MOV 31H,#60
19          CPL P1.0
20          AJMP LOOP
21          END

```

编译通过后,将 S16 文件夹中的 hex 文件烧录到 89C51 芯片中,将芯片插入到 S1 型 LED 试验板上,接上 5V 稳压电源,这时 P1.0 外接的 LED 熄灭;过 60 秒后,P1.0 外接的 LED 点亮;再过 60 秒后,LED 又熄灭;……如此循环不已。需说明的是,由于程序还要进行软件计数,占用了一定的时间,因此实际的定时时间要稍大于 1 分钟,这可以通过调整定时初值而得到精确的时

间。

我们对程序进行分析解释。

序号 1(程序解释,以下同):程序开始。

序号 2:跳转到 MAIN 主程序处。

序号 3:主程序 MAIN 从地址 0030H 开始。

序号 4:30H 单元置常数 20。

序号 5:31H 单元置常数 60。

序号 6:定时器 T1 方式 1。

序号 7、8:置定时初值。

序号 9:禁止中断。

序号 10:启动定时器 T1。

序号 11、12:查询 T1 的溢出标志 TF1。TF1=0 定时未到,转 LOOP 继续查询;TF1=1 定时到,转 LOOP1。

序号 13、14:重载定时初值。

序号 15:判是否到 1 秒?未到 1 秒,转 LOOP 继续循环;到 1 秒,向下执行。

序号 16:30H 单元重置常数 20。

序号 17:判是否到 1 分钟?未到 1 分钟,转 LOOP 继续循环;到 1 分钟,向下执行。

序号 18:31H 单元重置常数 60。

序号 19:P1.0 取反。

序号 20:跳转到 LOOP 处循环。

序号 21:程序结束。

(下一讲学习 MCS-51 的中断系统及有关实验)。

**配文优惠邮购** (每次邮费保价费 12 元):Keil 51 Windows 集成开发环境(已汉化光盘,邮购代号:K1):46 元。TOP851 多功能编程器(邮购代号:B1):400 元。LED 输出试验板(邮购代号:S1):90 元。LED 数码管输出试验板(邮购代号:S2):140 元。5V 高稳定专用稳压电源(邮购代号:D1):35 元。邮购时只需在附言栏中写明邮购代号及数量并附上联系电话即可。

邮购地址:201103 上海市闵行区莲花路 2151 弄 57 号 201 室

联系人:吕超亚

电话:021-64066571 13044152947

技术支持 E-mail:zxh2151@sohu.com